
Abstract

This article describes the implementation of a rudimentary voltmeter
using a PIC16F874 microcontroller. The input voltage must be in the
range 0 V to 5 V. The program displays the voltage either in raw form
using the 10 bits provided by the analog-to-digital conversion withint the
PIC16F874 or in binary-coded decimal form.

Figure 1: Schematic drawing of a PIC16F874 programmed to function like a
voltmeter.

1 Introduction

This project describes the use of a PIC16F874 implementation to implement a
simple voltmeter. Figure 1 shows the schematic drawing for this project and
Listing 1 on page 10 contains the source code of the program.

The schematic drawing shows that a potentiometer provides analog voltage
signals in the range 0 V to 5 V. The wiper is connected to the PIC16F874’s AN0
pin, analog channel 0. Internally the PIC16F874 converts the sampled voltage
to a 10-bit digital representation. The digital value can be displayed on the
eight-bit LED display in either of two ways:

• In raw binary form or

• Using three binary-coded decimal (BCD) digits, representing volts, tenths
of volts, and hundredths of volts.

Because eight bits suffices for neither of these directly, the display is multiplexed.
There are three phases in the display of each converted value. Table 1 shows
the manner and order in which values are displayed. The processor displays a
new step every 500 ms, giving the display a blinking appearance with a pause
between successive displays. A new sample is taken at the start of every 1.5 s
sequence.

Step # Raw Data Display BCD Display

0 Blank Blank
1 8 most significant bits Right-justified BCD digit

for volts
2 2 least significant bits, left

justified
BCD digits for tenths and
hundredths of volts

Table 1: Multiplexing the LED display

2 Analog-to-digital Conversion

The internal voltage references are used for this project. It is assumed that the
lower reference is V −

REF = 0 V and the upper reference is V +
REF = 5 V exactly.

If this assumption is violated, the voltmeter will not yield accurate results. To
get more accurate results, a precision voltmeter could be used.

The general expression for converting from an analog voltage x to a 10-bit
digital representation n using the PIC16F874 is given by

n =

⌊(
x− V −

REF

V +
REF − V −

REF

)
210

⌋
(1)

n =

⌊( x

5 V

)
210

⌋
. (2)

This is correct when the converted value is right justified in a 16-bit field, spread
across the two registers ADRESH and ADRESL. However, the value can also be left
justified in a 16-bit field, spread across the same two registers. In this format,
the representation is

n =

⌊( x

5 V

)
210

⌋
26. (3)

Our program needs to reverse the calculation in (1), however, whenever it
attempts to convert the raw binary data into BCD format. This can be done
mathematically by

x = V −
REF +

( n

210

) (
V +

REF − V −
REF

)
(4)

x =
( n

210

)
(5 V). (5)

Applying this to (3) we get

x =
( n

216

)
(5 V) =

(
5 V
216

)
(n). (6)

The first of these forms expresses x as some fraction of the distance from 0 V
to 5 V; the second expresses it as some multiple of the basic stepsize, 5 V

216 =
76.29 µV.1 The first therefore offers us the opportunity to perform integer
multiplication of n by 5 V. The latter requires we multiply n by a rather more
involved quantity, 76.29 µV.

Integer multiplication by 510 = 1012 is easy in a computer. All it requires is
to double the number, double it again, and add it to the original quantity. Mul-
tiplication by 76.2910 is more complicated and requires the use of a generalized
multiplication routine. The value can either be regarded as a real number—
necessitating the use of floating-point arithmetic—or it can be regarded as an
integral number of, say, hundreds of nanovolts:

76310 × 100 nV = 10 1111 10112 × 100 nV,

permitting use of integer arithmetic. For greater accuracy, it can instead be
regarded as an integral number of tens of nanovolts:

762910 × 10 nV = 1 1101 1100 11012 × 100 nV.

In this project we choose to do integer multiplication by 5 for simplicity.
A binary number n = bm−1 . . . b2b1b0 means

n = bm−1 . . . b2b1b0 (7)

= (bm−12m−1) + · · ·+ (b222) + (b121) + (b020). (8)

When this is divided by 2m, the result is

n

2m
=

bm−1 . . . b2b1b0

2m
= 0.bm−1 . . . b2b1b0. (9)

We can place the multiplier 5 in one byte and the multiplier n in a two-
byte field with the binary point implicitly placed at its left end. The result of
the multiplication will be three bytes with the binary point coming after the
leftmost byte.

Figure 2 shows pictorially how multiplication of a 16-bit quantity by the
constants 5 and 10 should proceed.

To get the decimal equivalent of the converted number of volts, therefore,
all we need to do is the following:

1. Multiply n by 5. The most significant digit is found in the leftmost byte
of the result.

2. Zero the most significant byte. This amounts to removing the integer
portion and leaving the fractional portion.

1The PIC16F874 uses 10 bits to represent the converted voltage. Because they are left-
justified in a 16-bit field, though, it is appropriate to use 216 rather than 210 in the denomi-
nator.

Figure 2: Multiplication of a 10-bit quantity by the constant 5 when left jus-
tified in a 16-bit field. An additional left shift converts this into the result of
multiplication by the constant 10 instead.

3. Multiply this by 10. The most significant digit is found in the leftmost
byte of the result.

4. Continue zeroing the most significant byte and multiplying by 10 to extract
as many significant digits as you wish. Since the PIC16F874 only provides
10 bits of accuracy, there is no point in extracting more than 3 digits in
all.

Note that unless the PIC16F874 encounters a whole voltage step, the digital
output will not change to the next step. Also, to get better results requires
rounding off, not rounding down. Rounding off can be accomplished by adding
half of the least significant amount of interest before doing all the manipulations.
For example, if we elect to extract three significant figures, we should add the
equivalent of 0.005 V first. To illustrate that this is right, consider how to round
2.647 V to the nearest 10 mV. We should round up:

Round(2.647, 2) =

⌊
102(2.647 + 0.005)

⌋

102

= 2.65.

The 16-bit digital equivalent of 0.005 V is

0.005 V
5 V

216 = 65.5 ≈ 6610 = 4216. (10)

So here is the revised algorithm:

1. Sample the analog voltage and left justify the result.

2. Add 004216 in preparation for later rounding.

3. Multiply by 5. The leftmost of the three resultant bytes contains the
number of whole volts.

4. Zero the most significant byte.

5. Multiply by 10. The leftmost of the three resultant bytes contains the
number of tenths of volts.

6. Zero the most significant byte.

7. Multiply by 10. The leftmost of the three resultant bytes contains the
number of hundredths of volts.

3 The main Program

The main program starts at line 417 of Listing 1. It initializes all data values
in lines 417–469 . Once it has accomplished this, it commences an endless task
of keeping the display up to date in lines 472–469.

3.1 Initialization

In lines 420–426 Port B is initialized for output by placing TRISB Init (eight
0 bits) into the TRISB register.

In lines 428–434 Port A is initialized for output by placing TRISA Init into
the TRISA register. This value is established in line 126 to ensure that bit 0
provides the analog input and bit 1 provides the user’s choice for the display
format.

The program initializes Timer 2 in lines 437–453. The initialization constants
are created in lines 106, 114, and 115. Timer 2 will interrupt the processor every
10 ms.

The program initializes the analog-to-digital converter in lines 455–463 with
constants created in lines 119 and 123. Analog values from input bit AN0 will
be converted to 10-bit values left-justified in registers ADRESH AND ADRESL.

Finally the display is blanked in lines 465–469.

3.2 Updating the Display

There are two display schemes, as mentioned in the Introduction. Bit 1 of Port
A is examined in line 475.

• If it is a 0, the display uses raw data.

• If it is a 1, the display uses binary-coded decimal (BCD).

Irrespective of the display format, the display is presented in three phases:

1. It is blank for 500 ms;

2. It shows the most significant 8 bits of the 10-bit converted value or the
number of whole volts in BCD for 500 ms;

3. It shows the least significant 2 bits of the 10-bit converted value or the
number of tenths and hundredths of volts in BCD for 500 ms;

To effect this, the main program relies upon the use of the variable DisplayNum,
maintained by the Timer 2 interrupt handler. It uses this variable in one of two
different table look-up schemes:

• The main program calls subroutine GetDisplayValue to display raw
data, starting at line 369 of Listing 1.

• It calls subroutine GetDisplayBCDValue to display BCD data, starting
at line 387 of Listing 1.

Both of these routines allocate two programs instructions for each table entry.
The first of these transfers the data to be displayed into register W; the second
returns it to the main program.

4 Interrupt Service Routines

4.1 Timer 2 Interrupt Handler

The Timer 2 interrupt handler only needs to do three things:

1. Count the number of occurrences of the Timer 2 interrupt. Since these
occur every 10 ms, the handler counts down from 50 to 0, after which
500 ms has elapsed.

2. Count the number of times 500 ms has passed. This entails counting down
from 3 to 0, after which 1.5 s has passed. At the end of this interval, the
handler initiates an analog-to-digital conversion.

3. Change the variable DisplayNum every 500 ms. This variable must stay
within the range [0, Displaymax− 1]. This permits the main program to
know what phase of the three-phase output should be displayed at any
time.

4.2 Analog-to-digital Converter Interrupt Handler

The analog-to-digital converter interrupts the processor whenever a conversion
is complete. The interrupt handler for these interrupts is in lines 212–367 of
Listing 1 and constitutes most of the code in the entire program.

Broadly speaking, the handler is responsible for putting the converted value
into a form suitable for output. If the desired output format is raw data, the
task is trivial. Registers ADRESH and ADRESL are copied to registers HighWord
and LowWord, respectively. The task is more involved when the desired output
format is BCD and details are in Section 2.

First, 4216 is added to the raw data in lines 231–243.
Then the number of whole volts is determined by multiplying the adjusted

data by 5. This is done in lines 245–280. The only peculiar aspect of the code is
the propagation of the carry bits. The incf instruction cannot be used because
it does not calculate the carry bit, whereas the addwf instruction does. But

when calculating the most significant byte, there is no possibility of an output
carry, as reference to Figure 2 on page 6 makes clear.

Next the number of tenths of volts is determined by zeroing the most sig-
nificant byte—the whole volts—and multiplying the rightmost two bytes by 10.
This is done in lines 282–322. The algorithm calls for multiplying by 5 and then
shifting the result one bit further to the left.

Finally the number of hundredths of volts is deterimined by once again ze-
roing the most significant byte—the tenths of volts—and multiplying the right-
most two bytes by 10. This is done in lines 324–364.

Listing 1: Voltmeter
;∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗
; This program provides a simple voltmeter to display the voltage ∗
; available at AN0. There are two displays, either one of which can∗
; be selected by a switch input: ∗

5 ; (1) if RA1 is a 1, the analog value is converted to BCD format. ∗
; (2) if RA1 is a 0, the analog value is displayed in raw format. ∗
; The analog data is provided at analog input AN0. It is sampled ∗
; once every 1.5 s . Eight LEDs provide the output. ∗
; ∗

10 ; When output is in BCD format, it is presented in three ∗
; phases: ∗
; (1) Blank for 0.5 s . ∗
; (2) A right− justified BCD representation of the number of volts ∗
; is displayed for 0.5 s . ∗

15 ; (3) Two BCD digits are displayed for 0.5 s. The leftmost of ∗
; these digits represents the number of tenths of volts ; ∗
; the rightmost represents the number of hundredths of volts. ∗
; ∗
; When output is in raw format, it is still presented in three ∗

20 ; phases: ∗
; (1) Blank for 0.5 s . ∗
; (2) Eight bits representing the most significant eight bits of ∗
; the 10−bit converted value. ∗
; (3) Two bits, left justified , representing the least significant ∗

25 ; two bits of the 10−bit converted value. ∗
;∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗
; ∗
; Filename: voltmeter.asm ∗
; Date: 11 March 2004 ∗

30 ; File Version: 1 ∗
; ∗
; Author: CDR Charles B. Cameron, USN ∗
; Company: United States Naval Academy ∗
; ∗

35 ; ∗
;∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗
; ∗
; Files required : ∗
; ∗

40 ; p16f874.inc ∗
; ∗
;∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗
; ∗
; Notes: ∗

45 ; It is assumed that the processor is driven by a 4 MHz crystal. ∗
; It is also assumed that Vdd = 5 V exactly. To the extent that ∗
; this assumption is in error , the converted voltages also will be ∗
; inaccurate. ∗
; ∗

50 ;∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗

list p=16f874 ; list directive to define processor
#include <p16f874.inc> ; processor specific variable definitions

55
; The following configuration choices turn code protection off , watch

; dog timer off ,brown−out reset disabled, power−up timer enabled, HS
; oscillator mode selected,flash program memory write disabled,
; low−voltage in−circuit serial programming disabled, and data EE

60 ; memory code protrection off.
ConfigBits set CP OFF & WDT OFF & BODEN OFF & PWRTE ON
ConfigBits set ConfigBits & HS OSC & WRT ENABLE ON
ConfigBits set ConfigBits & LVP OFF & CPD OFF

CONFIG ConfigBits
65

;∗∗∗∗∗ VARIABLE DEFINITIONS
w temp EQU 0x20 ; variable used for context saving

; Addresses A0, 120, and 1A0 also
70 ; are set aside for this purpose.

status temp EQU 0x21 ; variable used for context saving

cblock 0x22
temp ; A location for temporary values

75 count ; Used to count 10 ms interrupts.
ADC count ; Used to counts 500 ms intervals.
HighWord ; High word of the ADC conversion.
LowWord ; Low word of the ADC conversion.
DisplayNum ; Indicates which display is active .

80 ; Valid values are:
; 0 = blank display
; 1 = display most significant byte
; 2 = display least significant byte

y0 ; y = y2:y1:y0, a three−byte number
85 y1

y2
z0 ; z = z2:z1:z0, a three−byte number
z1
z2

90 OnesDigit ; BCD representation of analog volts
TenthsDigit ; BCD representation of analog deci volts
HundredthsDigit ; BCD representation of analog centi volts

endc

95 DisplayBlank equ 0
DisplayHigh equ 1
DisplayLow equ 2
DisplayMax equ 3 ; If the DisplayNum reaches this value, reset it .

100 ; Port A bit definitions
; Bit 0: AN0 Input from a potentiometer
; Bit 1: RA1 1 if output is to be in BCD form
; 0 if output is to be in raw 10−bit binary form
DisplayPreference equ 1 ; points to bit 1 of Port A

105
count init equ D’50’ ; Count 50 x 10 ms interrupts for 500 ms.
ADC count init equ D’3’ ; Count 3 x 500 ms for 1.5 s

INTCON Init equ B’11000000’ ; Bit 7 = GIE, Bit 6 = PEIE
110 ; Assume f osc = 4 MHz, so T inst = 1 us

; If x = prescaler , y = PR2, and z = postscaler, we want
; x (y+1) (z+1) = 10,000 us = 10 ms = Timer 2 interrupt interval
; Choose x = 4, y = 249, z=9

PR2 Init equ D’250’−1 ; y = 249
115 T2CON Init equ B’01001001’ ; Bits 6..3 => z=1001=9; Bits 1..0=01 => x=4

TRISB Init equ B’00000000’ ; All Port B bits are for output.

ADCON0 Init equ B’01000001’ ; Bits 7,6=01=> T AD = 8 T OSC = 2 us
120 ; Bits 5..3 = 000 => Channel 0

; Bit 2 = 0 => Don’t start a conversion
; Bit 0 = 1 => Turn the ADC on & start acquistion

ADCON1 Init equ B’00001110’ ; Bit 7 = 0 => left justified result
; Bit 6..4 = xxx

125 ; Bit 3..0 = 1110 => AN0, other Port A bits digital
TRISA Init equ B’00000011’ ; Let all Port A pins be used for output,

; except bit 0 is for analog input.
; Also, bit 1 is used for digital input
; DisplayPreference

130
; Macro Bankset n takes a parameter n in the range 0 to 3, inclusive
; It sets the STATUS bits RP1 and RP0 to provide access to the specified
; bank, n.
BankSet macro n

135 if n==0 ; Bank 0
bcf STATUS,RP0
bcf STATUS,RP1

endif
if n==1 ; Bank 1

140 bsf STATUS,RP0
bcf STATUS,RP1

endif
if n==2 ; Bank 2

bcf STATUS,RP0
145 bsf STATUS,RP1

endif
if n==3 ; Bank 3

bsf STATUS,RP0
bsf STATUS,RP1

150 endif
endm

155 ;∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗
ORG 0x000 ; processor reset vector
clrf PCLATH ; ensure page bits are cleared
goto main ; go to beginning of program

160
ORG 0x004 ; interrupt vector location
movwf w temp ; save off current W register contents
movf STATUS,w ; move status register into W register
bcf STATUS,RP0 ; ensure file register bank set to 0

165 movwf status temp ; save off contents of STATUS register

; Two ISR handlers are provided, one for Timer 2 interrupts and the
; other for A/D converter interrupts.

170 btfsc PIR1,TMR2IF

call HandleTimer2
btfsc PIR1,ADIF
call HandleADC

175 bcf STATUS,RP0 ; ensure file register bank set to 0
movf status temp,w ; retrieve copy of STATUS register
movwf STATUS ; restore pre−isr STATUS register contents
swapf w temp,f
swapf w temp,w ; restore pre−isr W register contents

180 retfie ; return from interrupt

HandleTimer2
bcf PIR1,TMR2IF ; Clear the interrupt flag

185
; Decrement the Timer 2 interrupt counter. When it
; reaches zero, 500 ms has passed. Decrement the
; 500 ms counter.
decfsz count,F

190 goto HandleTimer2End
movlw count init ; Re−initialize the 10 ms counter.
movwf count
decfsz ADC count,F
goto UpdateDisplay

195 ; When the ADC count reaches zero, 1.5 s has passed so
; start an A/D conversion.
BSF ADCON0,GO DONE
movlw ADC count init ; Re−initialize the 500 ms counter.
movwf ADC count

200
UpdateDisplay

; Update the display every 500 ms
incf DisplayNum,F
movlw DisplayMax ; If DisplayNum = DisplayMax, zeroize it.

205 subwf DisplayNum,W
btfsc STATUS,Z
clrf DisplayNum

HandleTimer2End
210 return

HandleADC
bcf PIR1,ADIF ; Clear the interrupt
; Save the converted values

215 movf ADRESH,W
movwf HighWord
BankSet 1
movf ADRESL,W
BankSet 0

220 movwf LowWord

; If we’re displaying BCD digits,
; then we want to round the result to the
; nearest decivolt . To do this,

225 ; add 0.005 V, then truncate. In 16−bit binary,
; this is (0.005 V / 5 V)∗2ˆ16 = 66 = 42h.
; If we’re displaying raw data, skip

; to the end of the interrupt handler.
btfss PORTA,DisplayPreference

230 goto HandleADCEnd
RoundValues

clrf y2
clrf z2
movlw H’42’

235 addwf LowWord,F
btfss STATUS,C ; Add the carry into HighWord if C=1.
goto ConvertToBCD
movlw 1
addwf HighWord,F

240 btfsc STATUS,C
incf y2 ; Add the carry into y2 if C=1.
movf y2,W ; Since y2 = 0 initially , it ’s safe to use incf
movwf z2

245 ConvertToBCD
; Calculate volts , decivolts , and centivolts
; First calculate the whole volts digit by multiplying
; the left− justified A/D result by 5 since the
; voltage references are assumed to be 5 V and 0 V.

250 movf LowWord,W ; Let z = y = y2:y1:y0 <− x = 0:HighWord:LowWord
movwf y0
movwf z0
movf HighWord,W
movwf y1

255 movwf z1

bcf STATUS,C
rlf y0,F ; Shift y2:y1:y0 left 1 bit
rlf y1,F ; y = 2x

260 rlf y2,F
bcf STATUS,C
rlf y0,F ; Shift y2:y1:y0 left 1 bit
rlf y1,F ; y = 4x
rlf y2,F

265 movf z0,W
addwf y0,F
btfss STATUS,C
goto ComputeUnitsDigit y1
movlw 1 ; Propagate the carry into y1

270 addwf y1,F
btfsc STATUS,C
incf y2,F ; Propagate the carry into y2

ComputeUnitsDigit y1
movf z1,W

275 addwf y1,F ; y <− y + x = 4x + x = 5 x
btfsc STATUS,C
incf y2,F ; Propagate the carry into y2
movf z2,W
addwf y2,W

280 movwf OnesDigit

; Now compute the tenths digit.
clrf y2 ; Eliminate the whole digits , first
; At this point , y1:y0 contains fractional volts

285 ; Multipy this by 10 and it will contain tenths of volts plus fractional
; tenths of volts . Accomplish this by multiplication by 5, then doubling.
movf y0,W
movwf z0 ; z <− y
movf y1,W

290 movwf z1
movf y2,W
movwf z2
bcf STATUS,C
rlf z0,F

295 rlf z1,F
rlf z2,F ; z <− 2 y
bcf STATUS,C
rlf z0,F
rlf z1,F

300 rlf z2,F ; z <− 4 y
movf y0,W ; z <− z + 4y = 5y
addwf z0,F
btfss STATUS,C
goto ComputeTenthsDigit y1

305 movlw 1 ; Propagate the carry into y1
addwf z1,F
btfsc STATUS,C
incf z2,F ; Propagate the carry into y2

ComputeTenthsDigit y1
310 movf y1,W

addwf z1,F
btfsc STATUS,C
incf z2,F
movf y2,W

315 addwf z2,F
movf z2,W
bcf STATUS,C ; z <− 2 z = 10 y
rlf z0,F
rlf z1,F

320 rlf z2,F
movf z2,W
movwf TenthsDigit

; Now compute the hundredths digit
325 clrf z2 ; Eliminate the tenths , first .

; At this point , z1:z0 contains 10 x fractional volts .
; Multiply it by ten to isolate the hundredths of volts .
movf z0,W
movwf y0 ; z <− y

330 movf z1,W
movwf y1
movf z2,W
movwf y2
bcf STATUS,C

335 rlf z0,F
rlf z1,F
rlf z2,F ; z <− 2 y
bcf STATUS,C
rlf z0,F

340 rlf z1,F
rlf z2,F ; z <− 4 y

movf y0,W ; z <− z + 4y = 5y
addwf z0,F
btfss STATUS,C

345 goto ComputeHundredthsDigit y1
movlw 1
addwf z1,F ; Propagate the carry into y1
btfsc STATUS,C
incf z2,F ; Propagate the carry into y2

350
ComputeHundredthsDigit y1

movf y1,W
addwf z1,F
btfsc STATUS,C

355 incf z2,F
movf y2,W
addwf z2,F
movf z2,W
bcf STATUS,C ; z <− 2 z = 10 y

360 rlf z0,F
rlf z1,F
rlf z2,F
movf z2,W
movwf HundredthsDigit

365
HandleADCEnd

return

GetDisplayValue
370 ; This subroutine returns one of three things :

; (1) Zero, if W=0
; (2) The most significant 8 converted bits if W=1
; (3) The least significant 8 converted bits if W=0
bcf STATUS,C

375 movwf temp
; Doubling the index lets us put two instructions in
; each table location .
rlf temp,W
addwf PCL,F

380 clrw ; Index 0
return
movf HighWord,W ; Index 1
return
movf LowWord,W ; Index 2

385 return

GetDisplayBCDValue
; This subroutine returns one of three things :
; (1) Zero, if W=0

390 ; (2) The right− justified BCD units digit if W=1
; (3) Two BCD digits, for tenths and hundredths, if W=2
bcf STATUS,C
movwf temp
; Doubling the index lets us put two instructions in

395 ; each table location .
rlf temp,W
addwf PCL,F
clrw ; Index 0

return
400 movf OnesDigit,W ; Index 1

return
; Well, note that the index 1 code is more than just 2
; instructions . This is ok as long as we don’t need
; another index.

405 movf TenthsDigit,W ; Index 2
movwf temp
bcf STATUS,C
rlf temp,F ; Shift the BCD digit to
rlf temp,F ; positions 7 down to 4

410 rlf temp,F
rlf temp,F
movf HundredthsDigit,W ; and fold in the hundredths

; in bits 3 down to 0.
addwf temp,W

415 return

main

; ∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗∗
420 ; Here is the code to test a PIC16F874.

; Configure Port B for output. It will be used to drive 8 LED indicators.
clrf PORTB
movlw TRISB Init
BankSet 1

425 movwf TRISB
BankSet 0

; Configure Port A.
; a test signal .

430 clrf PORTA ; Initialize Port A by clearing the output latches .
BankSet 1
movlw TRISA Init
movwf TRISA
BankSet 0

435

; Configure Timer 2 to generate an interrupt every 10 ms.
movlw INTCON Init
movwf INTCON

440 bcf PIR1,TMR2IF
BankSet 1
bsf PIE1,TMR2IE
movlw PR2 Init
movwf PR2

445 BankSet 0
clrf TMR2
movlw T2CON Init
movwf T2CON
movlw count init

450 movwf count
movlw ADC count init
movwf ADC count
bsf T2CON,TMR2ON ; Turn Timer 2 on now.

455 ; Configure the ADC

bcf PIR1,ADIF
movlw ADCON0 Init
movwf ADCON0
BankSet 1

460 bsf PIE1,ADIE
movlw ADCON1 Init
movwf ADCON1
BankSet 0

465 ; Initialize the display with all blanks
clrf HighWord
clrf LowWord
movlw DisplayBlank
movwf DisplayNum

470

loop
; Check to see whether raw ADC data or BCD equivalents
; are needed

475 btfsc PORTA,DisplayPreference
goto DisplayBCDValues

DisplayBinaryValues
; Retrieve the correct binary display values and output them.

480 movf DisplayNum,W
call GetDisplayValue
movwf PORTB
goto loop ; Continue indefinitely .

485 DisplayBCDValues
; Retrieve the correct BCD display values and output them.
movf DisplayNum,W
call GetDisplayBCDValue
movwf PORTB

490 goto loop

END

Index

ADC, see analog-to-digital conver-
sion

analog, 3
analog-to-digital conversion, 8, 9

BCD, see binary-coded decimal
binary-coded decimal, 3, 4, 8

display, 3, 7

initialization, 7
interrupt, 8

LED, see light-emitting diode
light-emitting diode, 3

multiplication, 5

potentiometer, 3

table look-up, 8

voltage reference, 4
voltmeter, 3

